Build project for anisoplanatism…

0- COMPLETELY FINALISE INSTALLATION OF "CAOS LITE" BEFORE GOING ON !!

Then, within the CAOS interface…



1- Reproduce the project ``Anisoplanatism" here beside.

2- Click on the ATM module, its graphical user interface (GUI) opens, then change its parameters into your own ones (r_0, L_0, altitude of the layers, mainly), and finally save them with button ``Save".

3- Fix the parameters of the other modules.

4- Choose a value for the off-axis angle (typically in between 0" and 60") within second occurence of module SRC and, as a consequence, adapt the name of the saved PSFs within the two modules SAV (one for each module IMG, i.e. one for each considered wavelength: for example 500nm and 1650nm).

5- Run the simulation project by using button ``Run" within the CAOS interface (or with the IDL-CAOS command ``.rn ./Projects/Anisoplanatism/project.pro" for a project called ``Anisoplanatism").

6- Repeat steps 4 and 5 for each chosen value of the off-axis angle.

7- Compute the rms of the corrected wavefront and the FWHM for each resulting PSFs (two for each off-axis angle value) with the help of routine ``dataprocessing.pro".

(routine dataprocessing.pro — 1)

```
; dataprocessing.pro, revised in March 2025
; use: .rn ./Projects/Aniso_2025/dataprocessing (for a project named "Anisoplanatism")


; parameters to be fixed for each case
THETA    = '10'                        ; off-axis angle ["]
diam_tel = 1.                          ; telescope diameter [m]
n_real   = 100L                        ; nb of realizations
np       = 100L                        ; nb of x- and y-pixels for the wf
np1      = 60L                         ; nb of x- and y-pixels for img#1
np2      = 60L                         ; nb of x- and y-pixels for img#2

; wf data processing
wf=fltarr(np,np,n_real)                ; cube of wf
for i=1,n_real do begin
   restore, "./Projects/Aniso_2025/theta_"+THETA+"as/wf"+strtrim(i,2)+".sav"
   wf[*,*,i-1]=data.screen
endfor
pupil=data.pupil                       ; telescope pupil

rms=fltarr(n_real)                     ; vector of rms [m]
idx=where(pupil gt 0.5)                ; indexes of valid pixels in which calculate the rms
for i=0,n_real-1 do begin
   dummy=wf[*,*,i]
   dummy=moment(dummy[idx], SDEV=sigma)
   rms[i]=sigma
endfor
print, "mean rms=", mean(rms)*1E9, " nm"
```

(routine dataprocessing.pro — 2)

```
; 500-nm images processing
img500nm=fltarr(np1,np1,n_real)      ; cube of 500-nm PSFs
for i=1,n_real do begin
   restore, "./Projects/Aniso_2025/theta_"+THETA+"as/PSF500nm"+strtrim(i,2)+".sav"
   img500nm[*,*,i-1]=data.image
endfor

PSF_LE = total(img500nm,3)           ; long-exposure PSF
LAMBDA = data.lambda                 ; wavelength [m]
RES    = data.resolution             ; pixel size ["]
dummy  = gauss2dfit(PSF_LE,a) & sig = (a[3]+a[2])/2.
fwhm   = 2*sig*sqrt(2*alog(2))*RES ; FWHM ["]
print, "FWHM = ", fwhm, '" = ', fwhm/(LAMBDA/diam_tel*!RADEG*3600), " lambda/D"

; H-band images processing
imgHband=fltarr(np2,np2,n_real)      ; cube of H-band PSFs
for i=1,n_real do begin
   restore, "./Projects/Aniso_2025/theta_"+THETA+"as/PSF1650nm"+strtrim(i,2)+".sav"
   imgHband[*,*,i-1]=data.image
endfor

PSF_LE = total(imgHband,3)           ; long-exposure PSF
LAMBDA = data.lambda                 ; wavelength [m]
RES    = data.resolution             ; pixel size ["]
dummy  = gauss2dfit(PSF_LE,a) & sig = (a[3]+a[2])/2.
fwhm   = 2*sig*sqrt(2*alog(2))*RES ; FWHM ["]
print, "FWHM = ", fwhm, '" = ', fwhm/(LAMBDA/diam_tel*!RADEG*3600), " lambda/D"
```

(Another metrics: the Strehl ratio — 1)

$$S = \frac{I_{\text{post AO}}[0,0]}{I_{\text{perfect}}[0,0]}$$

where *I[0,0]* is the intensity of the PSF at the optical center of the field (K. Strehl, Zeit. Instrumenkde 22, 213 (1902)).

$$S \simeq \exp\{-\sigma^2_{\text{post AO}}\}$$

in the framework of the Maréchal's approximation, where the variance (in radians²) is supposed to be small enough...







(Another metrics: the Strehl ratio — 2)

- Approximation that neglects tip-tilt: ratio of the maxima ($S \approx max(I)/max(I_{ideal})$)

- Ratio of the values at the centre of the image ≈ ratio of the OTF (see for example the paper of Roberts et al.)

- From Tokovinin (PASP, 2002):

$$S = \frac{I_{max}}{I_{tot}} \frac{4}{\pi} \left(\frac{\lambda_{CCD}}{D \Delta x} \right)^2$$

```
----------------------------------------------
 Report "Imaging through turbulence" (M1 MAUCA)
----------------------------------------------


----------------------
- Preliminary measures      (individual)    [/10]
----------------------
+ introduction/context
+ PSD(r0, L0)
+ => influence of r0 and L0
+ rms(r0, L0)
+ => influence of r0 and L0
+ FWHM(r0 or lambda=>r0, L0)
+ => influence of r0 and L0
+ => comparison with the "seeing" lambda/r0
+ noisy images
+ any personal development ?



---------------------------
- Anisoplanatic error study (binomial)      [/10]
---------------------------
+ introduction on anisoplanatism
+ CAOS modeling brief description (+L/theta_max+Δx/NΔx)
+ wf measures: rms(theta) (+input rms)
+ => var_aniso proportional to theta^(5/3) ?
+ => Strehl(theta, lambda)
+ => ccl on the influence of theta and lambda
+ img measures: FWHM(theta, lambda)
+ => ccl on the influence of theta and lambda
+ any personal development ?
```